两辆平板车装货问题

有七种规格的包装箱要装到两辆平板车上.包装箱的宽和高是一样的，但厚度t (厘米)和重量w (公斤)是不同的。表1给出了每种包装箱的厚度，重量以及数量。

表 1　包装箱参数信息

包装箱类型	C_1	C_2	C_3	C_4	C_5	C_6	C_7
厚度 t(厘米)	48.7	52.0	61.3	72.0	48.7	52.0	64.0
重量 w(公斤)	2000	3000	1000	500	4000	2000	1000
件数	8	7	9	6	6	4	8

每辆平板车有10.2米的地方可用来装包装箱(象面包片那样)，载重为40吨。由于地区货运的限制，对C5,C6,C7类包装箱的总数有一个特别的限制：这三类包装箱所占空间(厚度)不能超过302.7厘米。

问题要求：设计一种装车方案，使剩余的空间最小。

分析与解答：

设 C_i 类包装箱厚度 t_i 厘米，重量 w_i 公斤，件数为 n_i 件。

设第一辆车装载 C_i 类包装箱 x_i 件，第二辆车装载 C_i 类包装箱 y_i 件 $(i=1,2,\cdots,7)$。

第一辆车剩余空间 $Z_1 = 1020 - \sum_{i=1}^{7} t_i . x_i$

第二辆车剩余空间 $Z_2 = 1020 - \sum_{i=1}^{7} t_i . y_i$

总剩余空间最小为目标，则目标函数为:

$$\min Z = Z_1 + Z_2$$

约束满足:

1). 件数满足 $x_i + y_i \le n_i$ $\quad i = 1,2,\cdots,7$，且 x_i, y_i 为整数。

2). 各辆车载重量过 40 吨，有：

$$\sum_{i=1}^{7} w_i.x_i \le 40000, \quad \sum_{i=1}^{7} w_i.y_i \le 40000$$

3). 各辆车载长度不超过 1020 厘米,有:

$$\sum_{i=1}^{7} t_i.x_i \le 1020, \quad \sum_{i=1}^{7} t_i.y_i \le 1020$$

4) C_5, C_6, C_7 类包装箱的厚度不能超过 302.7 厘米,有:

$$\sum_{i=5}^{7} t_i.(x_i + y_i) \le 302.7$$

总的线性规划模型：

$$\min Z=(1020-\sum_{i=1}^{7}t_i.x_i)+(1020-\sum_{i=1}^{7}t_i.y_i)$$

$$s.t.\begin{cases}\sum_{i=1}^{7}t_i.x_i\le 1020\\ \sum_{i=1}^{7}t_i.y_i\le 1020\\ x_i+y_i\le n_i \qquad i=1,2,\cdots,7\\ \sum_{i=1}^{7}w_i.x_i\le 40000\\ \sum_{i=1}^{7}w_i.y_i\le 40000\\ \sum_{i=5}^{7}t_i.(x_i+y_i)\le 302.7\\ x_i,y_i\text{为非负整数}\end{cases}$$

得到解为:

$$x_1=6, x_2=2, x_3=9, x_4=1, x_5=0, x_6=0, x_7=0,$$

$$y_1=2, y_2=5, y_3=0, y_4=5, y_5=3, y_6=3, y_7=0$$

目标值 $Z=0.6$.

其中第一辆平板车剩余空间 0.1 厘米; 第二辆平板车剩余空间 0.5 厘米;

LINGO程序为:

```
!两辆平板车装货问题AMCM88B;
model:
sets:
num/1..7/:w,t,n,x,y;
endsets
data:
```

```
t=48.7,52.0,61.3,72.0,48.7,52.0,64.0;
w=2000,3000,1000,500,4000,2000,1000;
n=8,7,9,6,6,4,8;
enddata
min=(1020-@sum(num:t*x))+(1020-@sum(num:t*y));
@sum(num:t*x)<=1020;
@sum(num:t*y)<=1020;
@sum(num:w*x)<=40000;
@sum(num:w*y)<=40000;
@for(num(i):x(i)+y(i)<=n(i));
@sum(num(i)|i#GE#5#AND#i#LE#7:(x(i)+y(i))*t(i))<=302.7;
@for(num:@GIN(x));
@for(num:@GIN(y));
end
```

谢 谢！